SHARP

Graphing Calculator

EL-9650/9400

Handbook Vol. 2

Programmes



EL-9650



EL-9400

Read this first

This handbook was produced for practical application of the SHARP EL-9650 and 9400 Graphing Calculator. Both calculators include a highly convenient programming function, which enables automatic processing of both simple and complex calculations any number of times.

We would like to express our deepest gratitude to all the teachers whose cooperation we received in editing this book. In order to produce a handbook which is more replete and useful to everyone, we would welcome any comments or ideas on excercises. If you wish to contribute to future editions, **use the attached blank sheet or contact us by e-mail : osksp@hsa.osa.sharp.co.jp (for Windows 95)** or **oskspm@hsb.osa.sharp.co.jp (for Macintosh)**. When sending the data by e-mail, please include relevant information such as the explanation of the programme, parameters used in the programme and the listing of the programme. Please note that the programmes you send us may be opened to the public at this home page site or in other Sharp publications.

Note: Certain problems can not be solved with the EL-9400 as indicated in contents.

1. Entering and Editing a Programme:

Programmes can be entered and edited either by pressing the calculator keys or by downloading from a PC. To download programmes from a PC, you will need the CE-LK1 PC link software (sold separately).

A. Using calculator keys

- Creating a new programme:
  1. Press **2nd F** **PRGM** to display the programme menu.
  2. Press **C** **ENTER** to select the new programme menu. (See right)
  3. Enter the program title, then press **ENTER**.
  4. Enter the programme.
  5. Press **2nd F** **QUIT** to finish programming.

- Editing a programme:
  1. Press **2nd F** **PRGM** to display the programme menu.
  2. Press **B** and choose the number of the programme you wish to edit.
  3. Press **2nd F** **QUIT** to finish editing.

(See right)

B. Downloading from PC

- Creating a new programme:
  1. Using the CE-LK 1, select the **Model Type** from the **Tools** menu and click on the same model as your calculator.(click on EL-9600 for EL-9650)
  2. Select **New** from the **File** menu.
  3. Enter a programme name in **Title:**.
  4. Enter a program. (For details on entering a programme, refer to the operation manual.) (See right)
- Programmes can also be downloaded from Sharp's website at **http://www.sharp.co.jp/sc/excite/calculator/text/class96.htm** instead of creating a new programme.



SHARP

- Sending programmes from a PC:
  1. Using the CE-LK1, select the **Communication Port** from the **Link** menu and click on the port to be used.
  2. Turn off the EL-9650/9400 and connect it to the PC.
  3. Turn on the EL-9650/9400
  4. Select **Send…** from the **Link** menu of the CE-LK1 (See right)
  5. Specify the kind of drive, folder, and file, then select the file to be sent from the file list, and click on the **Select** button.
  6. Click on the **OK** button.



Note : For further details refer to the manual.

2. Executing a programme:

1. Press 2nd F PRGM to display the execute menu.
2. Press A ENTER and choose the number of the programme you wish to execute. (See right)
3. Follow the instructions.

3. Deleting a programme:

Press 2nd F OPTION C and then choose 5 to select the programme to be deleted.

Note: Do not try to erase a programme by resetting all memories to the initial condition as programme data to be stored will also be deleted. Also, it is advised to use the CE-LK1 PC link software to back up any programmes not to be erased.





4. Using the keys:

Press 2nd F to use secondary functions (in yellow).
To select "$\sin^{-1}$": 2nd F sin^{-1} ➔ Displayed as follows: 2nd F $\sin^{-1}$

Press ALPHA to use the alphabet keys (in blue).
To select A: ALPHA A ➔ Displayed as follows: ALPHA A

Press 2nd F A-LOCK to continue input of blue letters.
To input ABC: ALPHA A ALPHA A ALPHA A or 2nd F A-LOCK A A A
(To return to the normal function, press ALPHA again.)

5. Troubleshooting:

Following is a list of error codes and error messages.

When errors occur, refer to pages 12, 254, or 27 of the manual.

Error code	Error message	Error content
01	Syntax	Syntax error in equation or programme
02	Calculate	Execution of a division using 0, calculation beyond calculation range, etc.
03	Nesting	Reservation of 14 or more numerical values or 32 or more functions during execution.
04	Invalid	Matrix definition error
05 for	Dimension	Inconsistency in the dimension of matrix during arithmetic of a matrix or dimension of list STAT calculation.
07	Invalid DIM	Size of list and matrix input for calculation exceeds calculation range.
08	Argument	Inconsistency in argument of the structured function
09	Data Type	Invalid data type used in calculation
11	No define	Undefined list or matrix
12	Domain	Argument definition outside of domain
13	Increment	Increment error
17	Stat Med	Med-Med law (statistic) error
20	No Argument	No argument entered
21	Not pair $\int$ dx	Equation definition ($\int$ and dx as a pair) for integral calculus does not follow syntax.
22	Not pair [ ]	Not paired with specified "[ ]"
23	Not pair ( )	Not paired with specified "( )"
24	Not pair { }	Not paired with specified "{ }"
32	No data	Data does not exist
33	Graph Type	Error in graph type setting
37	No title	No title entered
38	Too many obj	More than 30 objects selected
40	Lbl duplicate	Same label name is used more than once within a programme
41	Lbl undefined	Label is not defined for Goto or Gosub
42	Lbl over	More than 50 labels are used within a programme
43	Gosub stack	Nesting of more than 10 subroutine stacks
44	Line too long	One line of programme exceeds more than 160 characters
45	Can't return	Use of return command without jumping from subroutine
46	Strage full	Attempt to create a file exceeding 99 (delete unnecessary files)
47	Coord type	Invalid coordinate system for command
90	Memory over	Over memory capacity
99	System error	User memory space cannot be secured

6. Page Layout



☆When the mark * appears on the key:
- Same series of key strokes can be done with screen touch.
- Key operations may also be carried out with the cursor (not shown).

Note: This handbook is only an example of how to use programming function of the EL-9650. The layout may vary with each screen.

Contents

*only for EL-9650

Other books available:
Graphing Calculator EL-9400 TEACHER'S GUIDE
Graphing Calculator EL-9650/9400 Handbook Vol. 1 (Algebra)

SHARP

Heron's Formula

Use Heron's formula to find the area of a triangle when the sides (A,B,C) of the triangle are known.

Calculation

$$S = \sqrt{D\,(D - A)\,(D - B)\,(D - C)}$$

$$D = \frac{(A + B + C)}{2}$$



FLOWCHART



PROGRAMME LIST(REAL MODE)

Title : HERON

```
Print "Input LENGTH
Input A
Input B
Input C
(A+B+C)/2⇒D
√(D (D-A) (D-B) (D-C) )⇒S
Print "S =
Print S
End
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
A	value of side A	D	value of D
B	value of side B	S	area
C	value of side C		

Exercise

Find the area of a triangle when sides A, B and C are 20, 35 and 40cm respectively.

Step	Key Operation	Display (When using EL-9650)
1 Specify the programme mode. Select the title HERON.	2nd F PRGM A*	
2 Enter the values A, B and C.	2 0 ENTER 3 5 ENTER 4 0 ENTER	
(Display of area)		
3 The area is approximately 350cm².		

Calculating Tension

Use the law of sines to find the tension when a pole of weight W is suspended with two strings, and the strings are balanced with the angles from the vertical line A and B.

Calculation

$$\frac{T}{\sin B} = \frac{S}{\sin A} = \frac{W}{\sin (A+B)}$$

$$\therefore T = W\frac{\sin B}{\sin (A+B)}$$

$$\therefore S = W\frac{\sin A}{\sin (A+B)}$$

T, S : tension W : weight
A, B : angles (6 sexagesimal numbers)





FLOWCHART



Enter angles and weight A, B and W.

Denominator in law of sines calculated. C= sin (A + B)

Tensions T and S calculated.
T = W∗sin B/C
T = W∗sin A/C

Tensions T and S displayed.

PROGRAMME LIST(REAL MODE)

Title : TENSION

```
Print "Input ANGLE
Input A
Input B
Print "Input WEIGHT
Input W
sin (A+B)⇒C
W∗sin B/C⇒T
W∗sin A/C⇒S
Print "TENSION
Print "T=
Print T
Print "S=
Print S
End
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
A	angle A	S	tension S
B	angle B	T	tension T
C	sin(A+B)	W	weight

SHARP

Exercise

Calculate the tension assuming weight=40kg, angle A=30° 15' 20", and angle B=27° 45' 40". Enter the angles with sexagesimal numbers.

Set up condition: decimal point digit number in TAB 3 Mode, decimal point in Fix Mode, and angle unit in Deg Mode.

2nd F SET UP C* 2* D* 3* B* 1* CL

Step	Key Operation	Display (When using EL-9650)
1 Specify the programme mode. Select the title TENSION.	2nd F PRGM A*	TENSION Input ANGLE A=?
2 Enter the values of angles A and B.	3 0 • 1 5 2 0 ENTER 2 7 • 4 5 4 0 ENTER	Input ANGLE A= 30.1520 B= 27.4540 Input WEIGHT W=?
3 Enter the value of weight.	4 0 ENTER	Input WEIGHT W= 40 TENSION T= 21.840 S= 23.795
4 Tension T is 21.840kg and S is 23.795kg.		

Involute (Inverse Involute)

Use the involute function for calculating gears etc. to find the angle of obliquity from the initial value and involute value.
Conversely, calculate the involute value from the angle of obliquity.

Calculation

Involute function : inv θ = tan θ - θ[rad]
Use Newton's method to find the inverse involute:

$$\theta_{i+1} = \theta_i - \frac{f'(\theta)}{f(\theta_i)} = \theta_i - \frac{\tan\theta_i - \theta_i - a}{\tan^2\theta_i}$$

$f(\theta) = a - \text{inv}\theta$

SP : involute curve
S : involute starting point
θ : angle of obliquity of point P



FLOWCHART



Enter 1 or 2.

To calculation of involute

To inverse involute calculation
Returns to START if entry neither 1 nor 2.

Calculation of involute.
Enter initial value and involute value.

Angle of obliquity calculated.

Judgment on repetition of calculation of angle of obliquity.

Calculation of inverse involute. Enter angle of obliquity.

Involute value calculated.
Involute value displayed.

PROGRAMME LIST (REAL MODE)

Title : INVOLUTE

```
Label START
ClrT
Print "SELECT 1/2
Input S
If S=1 Goto ANGLE
If S=2 Goto INVO
Goto START
Label ANGLE
Print "Input BEGIN
Input B
B ⇒ Z
Print "Input INVO
Input I
I ⇒ J
Label CALPRESS
tan Z ⇒ T
π*Z/180.0 ⇒ R
(T-R-J)/T^2 ⇒ D
180.0*(R-D)/π ⇒ Z
If int (10^8*D)≠0 Goto CALPRESS
Z ⇒ A
Print "ANGLE
Print A
End
Label INVO
Print "Input ANGLE
Input A
A ⇒ θ
tanθ -π*θ/180 ⇒ I
Print "INVOLUTE
Print I
End
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
D, R, T, J	working variable for calculating	θ	angle of obliquity
S	selecting calculation type	I	involute value
	(S=1: involute calculation)	A	input and output of angle
	(S=2: inverse involute calculation)	B	input of initial value
Z	initial value, angle of obliquity		

Exercise

(1) Find the angle of obliquity when the involute value is 0.0050912 and the initial value is 10.

(2) Find the involute value when the angle of obliquity is 14.1.

Set up condition: angle unit in Deg Mode and decimal point in Float Pt Mode.

2nd F SET UP B* 1* C* 1* CL

Step	Key Operation	Display (When using EL-9650)
1 Specify the programme mode. Select the title INVOLUTE.	2nd F PRGM A*	
2 Select involute calculation.	1 ENTER	
3 Enter the initial value and the involute value. (Display of angle of obliquity)	1 0 ENTER 0 • 0 0 5 0 9 1 2 ENTER	
4 Select inverse involute calculation.	ENTER 2 ENTER	
5 Enter the value of the angle of obliquity. (Display of involute value)	1 4 • 1 ENTER	

Calculating Illuminance and Luminous Intensity

Enter the luminous intensity of luminous source, distance, the angle between the perpendicular line and light ray, to find the illuminance of the illuminated side.
Conversely, find the luminous intensity of the source from the illuminance of the illuminated side.

Calculation

$$i = \frac{l \cdot \cos\theta}{r^2} \qquad l = \frac{r^2 \cdot i}{\cos\theta}$$

l : luminous intensity [candela] i : illuminance [lux]
r : distance [m] θ : angle [°]



FLOWCHART



Enter 1 or 2.

To calculation of luminous intensity

To calculation of illuminance

Jumps to subroutine DISTANCE.

Enter illuminance or luminous intensity.

Illuminance or luminous intensity calculated.

Illuminance or luminous intensity displayed.

Subroutine for entry of distance and angle

Entry

Returns to calling program.

PROGRAM LIST (REAL MODE)

Title : CAND LUX

```
Label START
ClrT
Print "CANDELA=1 LUX=2
Print "SELECT 1/2
Input S
If S=1 Goto CANDELA
If S=2 Goto LUX
Goto START
Label CANDELA
Gosub DISTANCE
Print "Input LUX
Input L
L ⇒I
R²*I/cos θ⇒C
Print "CANDELA
Print C
End
Label LUX
Gosub DISTANCE
Print "Input CANDELA
Input C
C⇒K
K*cos θ /R² ⇒L
Print "LUX
Print L
End
Label DISTANCE
Print "Input DISTANCE
Input D
D⇒R
Print "Input ANGLE
Input A
A⇒θ
Return
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
I	illuminance of luminated side	θ	angle
K	luminous intensity of luminous source	A	input of angle
R	distance	L	input and calculating luminous intensity
S	selecting calculation type (S=1: calculation of luminous intensity) (S=2: calculation of illuminance)	D	input of distance
		C	input and calculating illuminance

Exercise

(1) Find the luminous intensity of the luminous source of distance 10m, angle 60˚ and illuminance 20 lux.
(2) Find the illuminance of the illuminated side of distance 10m, angle 60˚ and luminous intensity 4000 candela.

Set up condition: angle unit in Deg Mode and decimal point in Float Pt Mode.

[2nd F] [SET UP] [B]* [1]* [C]* [1]* [CL]

Step	***Key Operation***	***Display*** (When using EL-9650)
1 Specify the programme mode. Select the title CAND LUX.	[2nd F] [PRGM] [A]*	
2 Select calculation of luminous intensity.	[1] [ENTER]	
3 Enter the values of distance, angle, and illuminance. (Display of luminous intensity)	[1] [0] [ENTER] [6] [0] [ENTER] [2] [0] [ENTER]	
4 Select calculation of illuminance. Enter the values of distance, angle, and luminous intensity. (Display of illuminance)	[ENTER] [2] [ENTER] [1] [0] [ENTER] [6] [0] [ENTER] [4] [0] [0] [0] [ENTER]	

Calculating Simple Harmonic Oscillation

Enter period, amplitude and time to calculate displacement at specified time, acceleration, angular velocity, and velocity. Also, display the changes during the entered time period on a graph.

Calculation

angular velocity : $\omega = \frac{2\pi}{T}$ displacement : $x = A \times \sin(\omega t)$

acceleration : $a = -\omega^2 \times x$ velocity : $v = A \times \omega \times \cos(\omega t)$

A : amplitude
t : time [sec]
T : period [sec]
ω : angular velocity [rad/sec]



FLOWCHART



Angular velocity, displacement, acceleration and velocity calculated.

W = angular velocity
H = displacement
B = acceleration,
V = velocity

Calculation result of angular velocity, displacement, acceleration and velocity displayed.

Range set and graph displayed.
Function: Y = D∗sin (W∗X)
X is time increase.
Xmin ··· 0, Xmax ··· E, Xscl ··· E/10
Ymin ··· -D, Ymax ··· D, Yscl ··· D/5

Text and graph display cleared.

PROGRAMME LIST(REAL MODE)

Title : OSCILLAT

```
Print "Input PERIOD
Input P
P⇒F
Print "Input AMPLITUDE
Input A
A⇒D
Label CALC
Print "Input TIME
Input T
T⇒E
2∗π/F⇒W
D∗sin (W∗E)⇒H
-(W²)∗H⇒B
D∗W∗cos (W∗E)⇒V
Print "ANGULAR VELOCITY
Print W
Print "MAGNITUDE
Print H
Print "ACCELERATION
Print B
Print "VELOCITY
Print V
Wait
E/10⇒X scl
D/5 ⇒Y scl
0⇒Xmin:E⇒Xmax
-D⇒Ymin:D⇒Ymax
Draw D∗sin (W∗X)
Wait
ClrT
ClrG
Goto CALC
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
B	acceleration	A	input of amplitude
E	time	P	input of period
V	velocity	T	input of time
W	angle of velocity (ω)	D	amplitude
H	displacement	F	period
Xscl	x-axis scale	X	time increase
Yscl	y-axis scale		

Exercise

Calculate angular velocity, etc., using period π, amplitude 1 and time 3 seconds and display the changes on a graph.

Set up condition: angle unit in Rad Mode and decimal point in Float Pt Mode.

[2nd F] [SET UP] [B]* [2]* [C]* [1]* [CL]

Step	*Key Operation*	*Display* (When using EL-9650)
1 Specify the programme mode. Select the title OSCILLAT.	[2nd F] [PRGM] [A]*	
2 Enter the values of period, amplitude, and time.	[2nd F] [π] [ENTER] [1] [ENTER] [3]	
3 (Display of angular velocity) (Display of displacement) (Display of acceleration) (Display of velocity)	[ENTER]	
4 (Display of graph of simple harmonic oscillation)	[ENTER]	
5	[ENTER]	

Electric Power Consumed on an AC Circuit

Enter the voltage effective value, frequency and resistance value to find the power value of the circuit with resistance R. Draw a graph of the changes in power over a period of time.

Calculation

P : power consumption I : effective value of current

V : effective value of voltage

$I_0 = N \cdot \sin \omega \cdot t$ $V_0 = M \cdot \sin \omega \cdot t$ $P_0 = I_0 \cdot V_0$

P_0 : change in amount of power with time

I_0 : change in amount of current with time

V_0: change in amount of voltage with time

N: maximum value of current M: maximum value of voltage

ω: angular velocity ($2 \pi S$) t : time S : frequency



FLOWCHART



Enter data (resistance, voltage and frequency).

Power calculated.
W = angular velocity
M = maximum voltage
N = maximum current
I = effective value of current
Z = power

Range for graph calculated.
Xmax, Xscl, Ymax, Yscl

Power displayed. (value of Z)

Function: $Y = N * M * (\sin (W * X))^2$

PROGRAM LIST(REAL MODE)

Title : AC POWER

```
Print "Input RESISTANCE
Input R
Print "Input VOLTAGE
Input V
Print "Input FREQUENCY
Input F
R⇒T
V⇒D
F ⇒ S
2∗π∗S ⇒W
D∗√2⇒ M
M/T⇒N
N/√2⇒I
D∗I⇒ Z
1/S ⇒ Xmax
Xmax/10⇒Xscl
N∗M⇒ Ymax
Ymax/10⇒Yscl
Print "WATT=
Print Z
Wait
0⇒Xmin
0⇒Ymin
Draw N∗M∗(sin (W∗X))²
End
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
S	frequency	Xscl	scale of x-axis
I	effective value of current	Ymax	maximum value of y-axis
T	resistance value	Yscl	scale of y-axis
D	effective value of voltage	V	input of voltage
W	angular velocity	R	input of resistance value
N	maximum value of current	F	input of frequency
M	maximum value of voltage	Z	value of power
Xmax	maximum value of x-axis		

Exercise

Find the power value of an AC circuit with resistance value 150Ω, voltage effective value 100V and frequency 50Hz and display on a graph the changes in power over a period of time.

Set up condition: angle unit in Rad Mode and decimal point in Float Pt Mode.

2nd F SET UP B * 2 * C * 1 * CL

Step	Key Operation	Display (When using EL-9650)
1 Specify the programme mode. Select the title AC POWER.	2nd F PRGM A *	
2 Enter the resistance value, voltage effective value, and frequency.	1 5 0 ENTER 1 0 0 ENTER 5 0 ENTER	
(Display of value power)		
3	ENTER	
(Display of graph)		

Angle of Vector

Use the matrix operation feature to find the angle θ which forms the standard vector and vector. The angle can be calculated at one time against the multiple vectors.

Calculation

Calculating vector inner product $\vec{a}\cdot\vec{b} = |\vec{a}||\vec{b}|\cos\theta$

Use the above expression to derive the following expression

$$\theta = \cos^{-1}\frac{\vec{a}\cdot\vec{b}}{|\vec{a}||\vec{b}|}$$

FLOWCHART



PROGRAMME LIST(MATRIX MODE)

Title : VECTOR

```
0⇒ I
0⇒K
Print " Input NUMBER
Input N
N⇒M
{M,2}⇒dim  (mat A)
{2,1} ⇒dim  (mat B)
{M,1}⇒dim  (mat C)
Label DATA
K + 1⇒K
Print " Input VECTOR
Print K
Input X
X⇒mat  A(K,1)
Input Y
Y⇒mat  A(K,2)
If  K<M Goto DATA
Print "Input FUNDAMENTAL VECTOR
Input  X
X⇒mat   B(1,1)
Input  Y
Y⇒mat  B(2,1)
√ (mat  B(1,1)²+mat  B(2,1)²)⇒B
mat  A∗mat  B⇒mat  C
Label  CALC
I + 1⇒I
If I>M  Goto  END
√ (mat  A(I,1)²+mat  A(I,2)²)⇒A
cos⁻¹  (mat  C(I,1) / (A∗B))⇒θ
Print "ANGLE OF VECTOR
Print I
Print "θ=
Print θ
Wait
Goto CALC
Label END
End
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
A	vector scalar quantity	θ	vector angle
B	standard vector scalar quantity	K	display
I	calculating counter	N	input of number of vectors
K	input counter	mat A	vector components
M	number of vectors	mat B	standard vector components
X	input of x component	mat C	vector inner product
Y	input of y component		

Exercise

Calculate the angle formed by the following 3 vectors and standard vector (2,3).
vector 1 (5, 8)
vector 2 (7, 4)
vector 3 (9, 2)

Set up condition: angle unit in Deg mode, and decimal point in Float Pt mode.

[2nd F] [SET UP] [B]* [1]* [C]* [1]* [CL]

Step		*Key Operation*	*Display*
1	Specify the programme mode. Select the title VECTOR.	[2nd F] [PRGM] [A]*	VECTOR Input NUMBER N=?
2	Enter the number of vectors.	[3] [ENTER]	VECTOR Input NUMBER N= 3 Input VECTOR 1 X=?
3	Enter the values of vector 1.	[5] [ENTER] [8] [ENTER]	1 X= 5 Y= 8 Input VECTOR 2 X=?
4	Enter the values of vectors 2 and 3.	[7] [ENTER] [4] [ENTER] [9] [ENTER] [2] [ENTER]	3 X= 9 Y= 2 Input FUNDAMENTAL VECTOR X=?
5	Enter the value of standard vector. (Display of angle of vector 1)	[2] [ENTER] [3] [ENTER]	X= 2 Y= 3 ANGLE OF VECTOR 1 θ= 1.684684318
6	(Display of angle of vector 2) (Display of angle of vector 3)	[ENTER] [ENTER]	ANGLE OF VECTOR 2 θ= 26.56505118 ANGLE OF VECTOR 3 θ= 43.78112476

Linear Transformation

Use the matrix to find four types of the linear transformation of x-axis symmetric transformation, y-axis symmetric transformation, similar transformation and revolution around the origin.

Calculation

1. Symmetric transformation to x-axis (Case 1)

$$\begin{pmatrix} X' \\ Y' \end{pmatrix} = \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix} \begin{pmatrix} X \\ Y \end{pmatrix}$$

2. Symmetric transformation to y-axis (Case 2)

$$\begin{pmatrix} X' \\ Y' \end{pmatrix} = \begin{pmatrix} -1 & 0 \\ 0 & 1 \end{pmatrix} \begin{pmatrix} X \\ Y \end{pmatrix}$$

3. Similar transformation with ratio of similitude K around origin (Case 3)

$$\begin{pmatrix} X' \\ Y' \end{pmatrix} = \begin{pmatrix} K & 0 \\ 0 & K \end{pmatrix} \begin{pmatrix} X \\ Y \end{pmatrix}$$

4. Transformation revolving around only angle B at the origin (Case 4)

$$\begin{pmatrix} X' \\ Y' \end{pmatrix} = \begin{pmatrix} \cos B & -\sin B \\ \sin B & \cos B \end{pmatrix} \begin{pmatrix} X \\ Y \end{pmatrix}$$

FLOWCHART



Declare array size, etc.
matH(2,2), matD(2,1), matA(2,1)
Enter coordinates before transformation.

Type of transformation specified with no from 1 to 4.

Jumps to destination corresponding to entered number.

XSYMMETRY
Data set of x-axis symmetric transformation
matH(1,1) = 1, matH(1,2) = 0,
matH(2,1) = 0, matH(2,2) = -1

YSYMMETRY
Data set of y-axis symmetric transformation
matH(1,1) = -1, matH(1,2) = 0,
matH(2,1) = 0, matH(2,2) = 1

SIMRATIO
Data set of similar transformation
Entry of ratio of similitude (R)
matH(1,1) = K, matH(1,2) = 0,
matH(2,1) = 0, matH(2,2) = θ

Data set of transformation by revolving
Entry of angle (A)
matH(1,1) = cos B, matH(2,1) = sin B,
matH(1,2) = -sin B, matH(2,2) = cos B,

Matrix H multiplied by matrix D.

Coordinates displayed.

PROGRAMME LIST

(MATRIX MODE)

Title : LINE TRNS

```
{2, 2}⇒dim(mat H)
{2, 1}⇒dim(mat D)
{2, 1}⇒dim(mat A)
Print "Input POINT
Input X
Input Y
X⇒mat D(1, 1)
Y⇒mat D(2, 1)
Label TYPE
Print "SELECT 1/2/3/4
Input S
ClrT
If S=1 Goto XSYMMETRY
If S=2 Goto YSYMMETRY
If S=3 Goto SIMRATIO
If S=4 Goto ROTATE
GotoTYPE
Label XSYMMETRY
1⇒mat H(1, 1)
0⇒mat H(2, 1)
0⇒mat H(1, 2)
-1⇒mat H(2, 2)
Goto TRANS
Label YSYMMETRY
-1⇒mat H(1, 1)
0⇒mat H(2, 1)
0⇒mat H(1, 2)
1⇒mat H(2, 2)
Goto TRANS
Label SIMRATIO
Print "Input SIMILITUDE RATIO
Input R
R⇒K
K⇒mat H(1, 1)
0⇒mat H(2, 1)
0⇒mat H(1, 2)
θ⇒mat H(2, 2)
Goto TRANS
Label ROTATE
Print "Input ANGLE
Input A
A⇒B
cos B⇒mat H(1, 1)
sin B⇒mat H(2, 1)
-sin B⇒mat H(1, 2)
cos B⇒mat H(2, 2)
Label TRANS
mat H*mat D⇒mat A
Print "mat A(1, 1)
Print mat A(1, 1)
Print "mat A(2, 1)
Print mat A(2, 1)
End
```


SHARP

PARAMETERS

Name of parameter	Content	Name of parameter	Content
B	angle	Y	y-coordinate
K	ratio of similitude	A	input of angle
S	selecting type	R	input of ratio of similitude
	(S=1: case 1, S=2: case 2,	mat A	coordinate after transformation
	S=3: case 3, S=4: case 4)	mat H	transformation data
X	x-coordinate	mat D	x,y-coordinate

Exercise

1. Transform symmetrically the point (3, 5) to the x-axis.
2. Rotate the point (2, 6) at 45° around the origin.

Set up condition: angle unit in Deg Mode and decimal point in Float Pt Mode.

[2nd F] [SET UP] [B] * [1] * [C] * [1] * [CL]

Step	Key Operation	Display
1 Specify the programme mode. Select the title LINE TRNS.	[2nd F] [PRGM] [A] *	LINE TRN Input POINT X=?
2 Enter the values of the point.	[3] [ENTER] [5] [ENTER]	Input POINT X= 3 Y= 5 SELECT 1/2/3/4 S=?
3 Select symmetric transformation to x-axis (case 1).	[1] [ENTER]	mat A(1,1) 3 mat A(2,1) -5
4 Select transformation revolving around only angle B at the origin (case 4).	[ENTER] [2] [ENTER] [6] [ENTER] [4] [ENTER]	Input POINT X= 2 Y= 6 SELECT 1/2/3/4 S= 4
5 Enter the angle value.	[4] [5] [ENTER]	Input ANGLE A= 45 mat A(1,1) -2.828427125 mat A(2,1) 5.656854249

Moving Average

Plot a moving average graph which helps to understand how the results change over a specified period. The progress of sales and amounts of consumption and production can also be seen.

Calculation

$$H_i = \frac{X_{i-(M-1)/2} + \ldots + X_i + \ldots X_{i+(M-1)/2}}{M}$$

$$\left(I = 1 + \frac{M-1}{2}, 2 + \frac{M-1}{2}, \ldots, n + \frac{M-1}{2} \right)$$

H_i : moving average
M : number of divisions
X_i : data
n : number of data

FLOWCHART



PROGRAMME LIST
(STAT MODE)

Title : MVIN AVG

```
Label MAIN
Print "Input DIVISION
Input D
D⇒M
1_Stats L1
If M≥n Goto MAIN
Rem RANGE
(xmax-xmin)/10⇒Yscl
0⇒Xmin
n⇒Xmax
1⇒Xscl
xmin⇒Ymin
xmax⇒Ymax
0⇒I
int (M/2)⇒K
Gosub COUNT
Label LOOP1
Gosub MOVINGSUM
If M≥J Goto LOOP1
Gosub AVERAGE
Label LOOP2
K⇒X
H⇒Y
Gosub COUNT
Label LOOP3
Gosub MOVINGSUM
If (I+M)>J Goto LOOP3
Gosub AVERAGE
Line (X, Y, K, H)
If K≠(n-int (M/2))Goto LOOP2
Wait
End
Label COUNT
I+1⇒I
I⇒J
0⇒S
Return
Label MOVINGSUM
S+L1(J)⇒S
J+1⇒J
Return
Label AVERAGE
S/M⇒H
K+1⇒K
Return
```

Parameters

name of parameter	content	name of parameter	content
H	moving average	S	moving sum
I	counter	X	starting point (x)
J	counter	Y	starting point (y)
K	counter	Yscl	scale of y-axis
M	number of divisions	B	input of number of divisions

Exercise

Find the moving average every three months (number of divisions: 3) from the following table of monthly sales.

Month	Jan.	Feb.	Mar.	Apr.	May.	Jun.	Jul.	Aug.
Sales[$]	300	326	323	344	300	401	398	450

On the graph, Xmax = 8, Ymin = 300, and Ymax = 450.

Set up condition: decimal point in Float Pt Mode.

[2nd F] [SET UP] [C]* [1]* [CL]

Step	*Key Operation*	*Display* (When using EL-9650)
1 Enter statistical data into L1.	[STAT] [A]* [ENTER]	
	[3] [0] [0] [ENTER] [3] [2] [6] [ENTER] [3] [2] [3] [ENTER] [3] [4] [4] [ENTER] [3] [0] [0] [ENTER] [4] [0] [1] [ENTER] [3] [9] [8] [ENTER] [4] [5] [0] [ENTER]	
2 Specify the programme mode. Select the title MVIN AVG.	[2nd F] [PRGM] [A]*	
3 Enter the number of divisions(3).	[3] [ENTER]	

Creating a Graph of Experimental Data

Graph the results of an experiment and examine the trends.
(Example: examined data relating to water vapour pressure and temperature.)

FLOWCHART



Enter statistical data using statistics feature before executing program.

Graph plotted using automatic scaling.

Counter

Data as coordinates (starting point and finishing point).

Line drawn between set coordinates.

Whether or not lines of no. of data drawn judged. Repeated until lines drawn by the no. of data.

PROGRAMME LIST(STAT MODE)

Title : XY GRAPH

```
ClrG
Rem DRAWING SD
2 -Stats L1,L2
Rem RANGE
xmin⇒Xmin
xmax ⇒ Xmax
ymin ⇒ Ymin
ymax⇒Ymax
(Xmax-Xmin) / 10⇒Xscl
(Ymax-Ymin) / 10⇒Yscl
Rem BROKEN LINE
0⇒I
Label  DRAWLOOP
I+1⇒ I
L1(I)⇒X
L2(I)⇒Y
L1(I+1)⇒Z
L2(I+1)⇒W
Line(X,Y,Z,W)
If (I+1) <n Goto DRAWLOOP
Wait
End
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
I	counter	Y	y of line starting point
X	x of line starting point	W	y of line finishing point
Z	x of line finishing point		

*n = number of statistical data

Exercise

The following table shows examined water vapour pressure. Draw a graph of this data.

Temperature [°C]	0	10	20	30	40	50	60	70	80	90	100
Pressure [mmHg]	4.581	9.205	17.532	31.826	55.339	92.558	149.47	223.79	355.29	525.90	760.00

Set up condition: decimal point in Float Pt Mode.

2nd F SET UP C * 1 * CL

Step	Key Operation	Display (When using EL-9650)
1 Enter statistical data into L1 and L2.	STAT A * ENTER	
2	0 ENTER 1 0 ... 1 0 0 ENTER (Other numbers not shown)	
3	4 • 5 8 1 ENTER ... 7 6 0 ENTER	
4 Specify the programme mode. Select the title XY GRAPH. (Drawing of graph)	2nd F PRGM A *	

Ordinary Differential Equations

Enter the initial conditions (X, Y) with the step H and interval T. Use Runge Kutta Gill method to solve the ordinary differential equation of first order.

Calculation

Use the following four steps of Runge Kutta Gill method to find the equation X_{n+1} and Y_{n-1} from X_n and Y_n. Input $Q_0 = 0$ at the starting point X_0.

1. $K_0 = Hf(X_n, Y_n)$, $R1 = (1/2)(K_0 - 2Q_0)$, $Y^{(1)} = Y_n + R_1$

2. $Q_1 = Q_0 + 3R_1 - (1/2)K_0$
   $K_1 = Hf(X_n + H/2, Y^{(1)})$, $R_2 = (1 - \sqrt{1/2})(K_1 - Q_1)$, $Y^{(2)} = Y^{(1)} + R_2$

3. $Q_2 = Q_1 + 3R_2 - (1 - \sqrt{1/2})K_1$
   $K_2 = Hf(Xn + H/2, Y^{(2)})$, $R_3 = (1 + \sqrt{1/2})(K_2 - Q_2)$, $Y^{(3)} = Y^{(2)} + R_3$

4. $Q_3 = Q_2 + 3R_3 - (1 + \sqrt{1/2})K_2$
   $K_3 = Hf(X_{n+1}, Y^{(3)})$, $R_4 = (1/6)(K_3 - 2Q_3)$, $Y_{n+1} = Y^{(3)} + R_4$
   $Q_4 = Q_3 + 3R_4 - (1/2)K_3$



FLOWCHART



Enter Data.
Initial coordinates (X, Y), step of x (H), and interval of solutions (T)

Data for calculation set.

Calculation executed.
Jumps to subroutine.

Jumps to subroutine.

Jumps to subroutine.

Jumps to subroutine.

Judgment of calculation end
If calculation result of I smaller than value of increase of I, calculation repeated again.

Following calculation performed when calculation result of x not equal to the value of increase of X.

$$P = \frac{(Z - S)(J - O)}{H} + O, \quad \begin{matrix} M = Z \\ N = P \end{matrix}$$

Prior processing for next calculation
$Z = Z + T$, $S = X$, $O = J$



Subroutine for calculating built-in function

f = -I ∗J
(Another equation can be used.)

PROGRAMME LIST(REAL MODE)

Title : RUNGE

```
Rem INITIAL
Print " Input X0
Input X
Print " Input Y0
Input Y
X⇒I
Y⇒J
Print " Input H
Input H
Print " Input T
Input T
1+√(2^-1)⇒A
1- √(2^-1)⇒B
I+T⇒Z
O⇒Q
I⇒S
Label MAIN
Rem 1
Gosub FORMULA
H∗F⇒K
(K-2∗Q) /2⇒R
J+R⇒J
Q+3∗R-K/2⇒Q
I+H/2⇒I
Rem 2
Gosub FORMULA
H∗F⇒K
B∗(K-Q)⇒ R
J+R⇒J
Q+3∗R-B∗K⇒Q
Rem 3
Gosub FORMULA
H∗F⇒K
A∗(K-Q)⇒ R
J+R⇒J
Q+3∗R - A∗K⇒Q
I+H/2⇒I
Rem 4
Gosub FORMULA
H∗F⇒K
(K - 2∗Q) /6⇒R
J+R⇒J
Q+3∗R - K/2⇒Q
If Z≤I Goto NEXT
I⇒S
J⇒O
Goto MAIN
Label NEXT
If Z≠I Goto SUB2
I⇒M
J⇒N
Label SUB1
ClrT
Print "XN=
Print M
Print "YN=
Print N
Wait
Z+T⇒Z
I⇒S
J⇒O
Goto MAIN
Label SUB2
(Z-S)∗(J-O) /H+O⇒P
Z⇒M
P⇒N
Goto SUB1
Label FORMULA
-I∗J⇒F
Return
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
A	value of $1+\sqrt{(1/2)}$	S	value of Xn-1
B	value of $1-\sqrt{(1/2)}$	T	interval
F	f (I,J)	I	Xn
H	step	J	Yn
K	calculating working area	Z	value of increase of X
O	value of Yn-1	X	input of X_0
P	increase of J	Y	input of Y_0
Q	value of Qn	M	indicates Xn
R	value of Rn	N	indicates Yn

Exercise

Initial settings: Y = 10 when X = 0. Find J when H = 0.01, T = 0.03 and I = 0.03, 0.06···. (The built-in differential equation is F = -I ∗ J.)

Set up condition: angle unit in Rad Mode and decimal point in Float Pt Mode.

2nd F SET UP B ∗ 2 ∗ C ∗ 1 ∗ CL

Step	***Key Operation***	***Display*** (When using EL-9650)
1 Specify the programme mode. Select the title RUNGE.	2nd F PRGM A ∗	RUNGE Input X0 X=?
2 Enter the values of X_0, Y_0, H and T.	0 ENTER 1 0 ENTER 0 • 0 1 ENTER 0 • 0 3 ENTER	RUNGE Input X0 X= 0 Input Y0 Y= 10 Input H H= 0.01 Input T T= 0.03
3 (Display of X1) (Display of Y1)	ENTER	XN= .03 YN= 9.995501012
4 (Display of X2) (Display of Y2)	ENTER	XN= .06 YN= 9.98201619
5 (Display of X3) (Display of Y3)	ENTER	XN= .09 YN= 9.959581902

Similar operation is performed hereafter.

Analysing with One-way Layout Method

Use the one-way layout method to verify whether there is a relation to the results achieved based on one condition. Analysis of variance is carried out with this method.

Calculation

Analysis of variance chart of one-way layout method

	Sum of squares (S)	Degree of freedom (θ)	Variance (V)	Variance ratio (F)
Factor	S_A = [A] - [X]	θ_A = A - 1	$V_A = S_A \div \theta_A$	$F_A = V_A \div V_E$
Error	S_E = [AS] - [A]	θ_E = A (N- 1)	$V_E = S_E \div \theta_E$	
Total	S_T = [AS] - [X]	θ_T = AN - 1		

$[X] = (\Sigma\,\Sigma\,Xij)^2 \div AN$
$[A] = \Sigma i\,(\Sigma j\,Xij)^2 \div N$
$[AS] = \Sigma i\,\Sigma j\,(Xij)^2$
A : number of levels
N: repeated frequency
X : number of data

FLOWCHART



PROGRAMME LIST(STAT MODE)

Title : VARIANCE

```
Rem INPUT
Print "Input LEVEL
Input L
L⇒A
Print "Input TIMES
Input T
T⇒N
0⇒W
0⇒B
0⇒C
1⇒S
Label LOOP2
N⇒dim(L1)
1⇒K
Label LOOP1
ClrT
S⇒L
K⇒T
Print "Input DATA
Print "LEVEL
Print L
Print "TIME
Print T
Input I
I⇒L1(K)
B+I⇒B
C+I²⇒C
K+1⇒K
If (N+1)>K Goto LOOP1
1_Stats L1
Σx⇒J
Print "Σx=
Print J
Wait
W+(Σx)²⇒W
S+1⇒S
If (A+1)>S Goto LOOP2
Rem CALCULATE
B²/A/N⇒X
W/N⇒Y
C⇒Z
Rem SUM OF SQUARES
Y-X⇒E
Z-Y⇒M
Z-X⇒P
Print "SUM OF SQUARES
Print E
Print "ERROR SUM OF SQUARES
Print M
Wait
Print "TOTAL SUM OF SQUARES
Print P
Wait
Rem DEGREES OF FREEDOM
A-1⇒Q
A*(N-1)⇒R
A*N-1⇒D
Print "DEGREES OF FREEDOM
Print Q
Print "DEGREES OF FREEDOM ──┐
Print R                      ↳ABOUT ERRORS
Wait
Print "SUM OF DEGREES OF FREEDOM
Print D
Wait
Rem VARIANCE
E/Q⇒V
M/R⇒U
Print "VARIANCE
Print V
Print "VARIANCE OF ERRORS
Print U
Wait
Rem VARIANCE RATIO
V/U⇒F
Print "VARIANCE RATIO
Print F
End
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
A	number of levels	V	variance factor
I	input of data	U	variance error
K	loop 1 counter	Y	$\Sigma i\ (\Sigma\ jxij)^2 / n$
J	indicating Σx	Q	degree of freedom factor
N	repeated frequency	R	degree of freedom error
S	loop 2 counter	D	degree of freedom total
X	$(\Sigma\Sigma\ xi)^2 / a / n$	T	input and indicating frequency
Z	$\Sigma i\ \Sigma j\ (xij)^2$	L	input and indicating number of levels
F	variance ratio factor	W	total sum of squares of each level
E	sum of squares factor	B	total sum (all data)
M	sum of squares error	C	stotal sum of squares (all data)
P	sum of squares total		

Exercise

When a mouse is given a dosage of hormone, the relationship between dosage amount and increase of mouse weight is as shown in the following table. Find the analysis of variance. If the value of the variance ratio is larger than the value of F- distribution table of the 5% level of significance, the relationship between the hormone amount and the increase of mouse weight is a causal relation.

	Increase mouse weight (grams/day)					
		10	20	30	40	50
Hormone (grams/mouse)	10	882	891	864	888	885
	20	923	915	923	912	930
	30	933	939	925	940	932

The number of levels (number of columns in the table) is A = 3
The repeated frequency (number of rows in the table) is N = 5

Set up condition: decimal point in Float Pt Mode.

[2nd F] [SET UP] [C]* [1]* [CL]

Step	Key Operation	Display (When using EL-9650)
1 Specify the programme mode. Select the title VARIANCE.	[2nd F] [PRGM] [A]*	VARIANCE Input LEVEL L=?
2 Enter the number of levels and the repeated frequency.	[3] [ENTER] [5]	VARIANCE Input LEVEL L= 3 Input TIMES T= 5
3	[ENTER]	Input DATA LEVEL 1 TIME 1 I=?

Step	Key Operation	Display (When using EL-9650)
4 Enter the statistical data in level 1.	8 8 2 ENTER 8 9 1 ENTER 8 6 4 ENTER 8 8 8 ENTER 8 8 5 ENTER	
(Display of total of hormone 10 g)		
5 Enter the statistical data in level 2.	ENTER 9 2 3 ENTER 9 1 5 ENTER 9 2 3 ENTER 9 1 2 ENTER 9 3 0 ENTER	
(Display of total of hormone 20 g)		
6 Enter the statistical data in level 3.	ENTER 9 3 3 ENTER 9 3 9 ENTER 9 2 5 ENTER 9 4 0 ENTER 9 3 2 ENTER	
(Display of total of hormone 30 g)		
7 (Display of sum of squares) (Display of error sum of squares)	ENTER	
8 (Display of sum of squares)	ENTER	
9 (Display of degrees of freedom) (Display of degrees of freedom about errors)	ENTER	
10 (Display of sum of degrees of freedom)	ENTER	
11 (Display of variance) (Display of variance of errors)	ENTER	
12 (Display of variance ratio)	ENTER	

The F-distribution chart shows that the value of F of upper probability P = 5% is 3.89. Since f > 3.98 in this example, the relationship between the hormone amount and the increase of mouse weight is a causal relation with 5% level of significance.

Calculating Parabolic Motion

Display on a graph the altitude change and the horizontal distance over a period of time when an object is thrown at initial velocity V_0 and angle θ, and find the horizontal distance and altitude after t seconds. Specify the angle in Deg.

Calculation

$X = V_0 \cdot \cos\theta \cdot T \qquad Y = V_0 \cdot \sin\theta \cdot T - \frac{1}{2} gT^2$

Initial velocity	V_0 [m/s]
Angle	θ [°]
Gravitational acceleration	g = 9.8 [m/s^2]
Time	T [s]



FLOWCHART



Enter velocity when thrown.

Highest altitude, throwing distance (horizontal distance), and time (duration of flight) in case of released angle 45˚ calculated and displayed.

Angle for throwing entered.

Entered angle less than or equal to 0˚ or larger than 90˚?

Highest altitude, throwing distance (horizontal distance), and time (duration of flight) for entered angle calculated and displayed.

Range of graph set based on values for released angle 45˚.

Graph (parabola) calculated and plotted.

Elapsed time counted.

Calculation and plotting repeated until D (time elapsed) reaches T (duration of flight).

Entered time less than or equal to 0 or more than T?

Altitude and distance after entered time elapses from throwing calculated and displayed.

Returns to entry of time.

PROGRAMME LIST(REAL MODE)

Title : PARABOLA

```
Print "V0 (M/S),θ,T(S)
Print "Input V0
Input V
2∗V∗sin 45/9.8⇒A
V²/9.8⇒B
V²/19.6⇒C
Print "HMAX=
Print C
Print "LMAX=
Print B
Print "TMAX=
Print A
Wait
Label THETA
Input θ
If θ ≤ 0 Goto THETA
If θ > 90 Goto THETA
V²∗(sin θ)²/19.6⇒H
V²∗sin (2θ)/9.8⇒L
2∗V∗sin θ/9.8⇒T
Print "H=
Print H
Print "L=
Print L
Print "T=
Print T
Wait
C/10⇒Yscl
B/10⇒Xscl
0⇒Xmin
0⇒Ymin
B⇒Xmax
C⇒Ymax
0⇒D
Label LOOP1
V∗cos θ∗D⇒X
V∗sin θ∗D-(0.5∗9.8∗D²)⇒Y
Pnt0N(X,Y)
D+(T/100)⇒D
If D≤T Goto LOOP1
Wait
Label TX
Print "Input TX
Input Z
If Z≤0 Goto THETA
If Z>T Goto THETA
V∗cos θ∗Z⇒X
V∗sin θ∗Z-(0.5∗9.8∗Z²)⇒Y
Print "X=
Print X
Print "Y=
Print Y
Wait
Line(0,Y,X,Y)
Line(X,0,X,Y)
Wait
Goto TX
0000
```

PARAMETERS

Name of parameter	Content	Name of parameter	Content
H	highest altitude	Xscl	scale of x-coordinate
L	horizontal distance	Z	input of time period
T	time	V	initial velocity (V_0)
X	distance (after time Z)	θ	angle (released angle)
Y	altitude (after time Z)	C	highest altitude when released at 90°
D	time elapsed	B	horizontal distance when released at 45°
Yscl	scale of y-coordinate	A	time period when released at 45°

Exercise

Find the horizontal distance and altitude three seconds after an object is thrown, when the initial velocity is 25m/sec and the angle is 52°.

Set up condition: angle unit in Deg mode, and decimal point in Float Pt mode.

[2nd F] [SET UP] [B] * [1] * [C] * [1] * [CL]

	Step	*Key Operation*	*Display* (When using EL-9650)
1	Specify the programme mode. Select the title PARABOLA.	[2nd F] [PRGM] [A] *	PARABOLA V0 (M/S),θ,T(S) Input V0 V=?
2	Enter the value of the initial velocity. (Highest altitude when released at 90°) (Distance when released at 45°) (Time when released at 45°)	[2] [5] [ENTER]	V= 25 HMAX= 31.8877551 LMAX= 63.7755102 TMAX= 3.607687659
3		[ENTER]	25 HMAX= 31.8877551 LMAX= 63.7755102 TMAX= 3.607687659 θ=?
4	Enter the angle value. (Display of highest altitude) (Display of horizontal distance) (Display of time until dropping of object)	[5] [2] [ENTER]	θ= 52 H= 19.80105063 L= 61.88110499 T= 4.020463029
5	(Display of graph of parabola)	[ENTER]	
6		[ENTER]	H= 19.80105063 L= 61.88110499 T= 4.020463029 Input TX Z=?
7	Enter the value of time period Z. (Display of distance after Z seconds) (Display of altitude after Z seconds)	[3] [ENTER]	4.020463029 Input TX Z= 3 X= 46.17461065 Y= 15.00080652
8	(Altitude and distance after Z seconds are displayed on the parabola graph.)	[ENTER]	

Key pad for the SHARP EL-9650 Calculator



1. Graphing keys
2. Power supply ON/OFF key
3. Alphabet specification key
4. Secondary function specification key
5. Display screen
6. Cursor movement keys
7. Clear/Quit key
8. Variable enter key
9. Calculation execute key
10. Communication port for peripheral devices

Key pad for the SHARP EL-9400 Calculator



1. Graphing keys
2. Power supply ON/OFF key
3. Alphabet specification key
4. Secondary function specification key
5. Display screen
6. Cursor movement keys
7. Clear/Quit key
8. Variable enter key
9. Calculation execute key
10. Communication port for peripheral devices

SHARP

Use this form to send us your contribution

Dear Sir/Madam

We would like to take this opportunity to invite you to create a mathematical problem which can be solved with the SHARP EL-9650 and 9400 graphing calculator, including the necessary procedures and definitions as outlined in the form below.

For this purpose, we would be grateful if you could complete the form and return it to us by fax or mail, specifying whether you have created the problem for the EL-9650 or the EL-9400. If your contribution is chosen, your name will be included in the next edition of The EL-9650/9400 Graphing Calculator Handbook or on our homepage. We regret that we are unable to return contributions. Also, please note that the problems you send us might be opened to the public at Sharp's home page.

We thank you for your cooperation in this project.

Name: (☐ Mr. ☐ Ms.) _____

School/College/Univ.: _____

Address: _____

_____ Post Code: _____ Country: _____

Phone: _____ Fax: _____

E-mail: _____

* You are making this sheet for the (☐ EL-9650, ☐ EL-9400).

SUBJECT: Write either a title or about the subject matter.

. . . . . .

INTRODUCTION and CALCULATION:
Write a brief explanation of the subject, and the formula with definitions, including a diagram if relevant.

. . . . . .

. . . . . .

. . . . . .

. . . . . .

. . . . . .

. . . . . .

. . . . . .

. . . . . .

. . . . . .

. . . . . .

PARAMETERS:

Define the parameters used in the programme.

PROGRAMME LIST:

List the procedure of data to be entered.

EXERCISE and SET UP CONDITION:

Include an example of a problem which can be solved with the formula. Write a step-by-step guide to solving the problem with an explanation.Detail any important conditions to be set up before solving the problem.

SHARP

SHARP CORPORATION Osaka, Japan
Fax:

SHARP
SHARP CORPORATION OSAKA,JAPAN